

## 財務計算あれこれ

第 6 回 財務計算におけるいろいろな例題 (1)  
 No. 6. Some Examples in Finance Problems (1)


(株) 竹内八ガネ商行      竹内寿一郎

## 1. はじめに

これまでの報告で、年利  $r$ 、年価・返済額  $M$  (Annuities・Payments)、および入出金の期間間隔について、全て一定であると仮定した場合<sup>【3】</sup>、<sup>【4】</sup>、<sup>【5】</sup>と、年利  $r$ 、年価・返済額  $M$  が一定でない場合<sup>【6】</sup>、および全ての項目が一定でない場合<sup>【7】</sup>について、基本関数を求めてきた。とくに後半のシリーズでは時間が十分に取れなかったこともあり、例題が幾分雑であったように思われる。そこでここではエクセルに組み込まれた関数を、J の関数で表現するにはどのようにしたらよいか、という観点から例題を取り上げ J で作った関数を実行してみた。紙数の関係で問題を 3 つに分け、(1) 一定の年価・年率で一定期間ごとにお金の出入りがある場合、(2) 年価・年率が一定ではなく、しかし一定期間ごとにお金の出入りがある場合、そして最後に (3) 期間が不定期のお金の出入りがある場合について纏めて例題を掲げて解説することにした。今回はそのうちの (1) について例をあげてみた。

## 2. 年価・金利が一定の場合の公式 (終価・現価・年価の公式)

例として上段が支払、下段が受取金額として、キャッシュフローを図示してみる。支払、受取が逆の場合は上下段を逆にすればよい。この図では基線から下の金額には分かりやすいようにマイナス表示してあるが、本来必要としないマイナスである。



ここで、期間は  $n$ 、 $M_0$  は初期支払額、 $M_n$  は満期支払額、各期の返済額または収益額  $M$ 、 $P$  は借入額または投資額、 $S$  は満期受取額、各期における変動利率を  $r$  とする

現価と終価

$$(1) \quad \begin{cases} S = (1+r)^n P \\ P = \frac{1}{(1+r)^n} S \end{cases}$$

## 期末と期首の問題

年価  $M$  が入ると、巷で言う「期末」とか「期首」が出てくるので式が複雑になる。以前から主張しているようにお金の出入りはその期の「終わり」か「始め」に行われるから期末、期首

というわけでは無い。英語でいうと期末は *Arrears*、期首は *Advance* であり、いわゆる「事後」か「事前」を意味する。ある時点で行われた入出金はその期分であるか、次の期分であるかの解釈の違いを言っているのであり、その時点がその期の「終わり」か「始め」であることに意味は無い。悪いことに英語 *Arrears*、*Advance* を期末、期首という間違った和訳を採用したために誤解が生じたのである。わたくし流に言えば、初期時点でお金が入り出す（通常頭金がある）場合が期首であり、満期にお金が入り出す場合が期末である。中間ではお金の入りは期末だろうと期首だろうと気にしない。厳密にいうと中間では期末ニアリーコール (*Nearly Equal*) 期首である。そこで、 $b$  を論理値からなるベクトル、 $b = (b_0, b_n)'$  として、期末: ( $M_0$  無,  $M_n$  有) = (0,1)、期首: ( $M_0$  有,  $M_n$  無) = (1,0)、無無: ( $M_0$  無,  $M_n$  無) = (0,0)、有有: ( $M_0$  有,  $M_n$  有) = (1,1)、のように定義することができる。このとき  $M_0 = M_n = M$  のときの級数の和の公式は

$$(2) \left\{ \begin{array}{l} S = b_n M + M(1+r) + M(1+r)^2 + \dots + M(1+r)^{n-1} + b_0 M(1+r)^n \\ \quad = \frac{(1+r)^{n+b_0} - (1+r)^{1-b_n}}{(1+r)^{n+b_0} - (1+\bar{b}_n r)} \cdot M = \frac{(1+r)^{n+b_0} - (1+\bar{b}_n r)}{(1+r)^{n+b_0} - (1+\bar{b}_n r)} \cdot M \\ P = b_0 M + \frac{M}{(1+r)} + \frac{M}{(1+r)^2} + \dots + \frac{M}{(1+r)^{n-1}} + b_n \frac{M}{(1+r)^n} \\ \quad = \frac{(1+r)^{n+b_0} - (1+\bar{b}_n r)}{r(1+r)^n} \cdot M \quad S = P(1+r)^n \text{ の関係を使った} \end{array} \right.$$

初項の関係で  $b_n$  が 0 と 1 (無 / 有) のときは和の式の中でそれぞれ  $(1+r)$ 、1 となり、最終項の関係で  $(1+r)^{n+b_0}$  での  $n+b_0$  が 0 と 1 (無 / 有) のときは和の式の中でそれぞれ  $n$ 、 $n+1$  となることが見てとれる。

これにオプション番号を付けるとすれば、エクセル<sub>[1]</sub>でも  $J_{[2]}$ でも期末が 0、期首が 1 であるからそれに合わせるようにし、それに加え、2 が無無、3 が有有とした。なお、 $(1+r)^{1-b_n}$  は *not*  $b_n = \bar{b}_n$  とおいて  $(1+\bar{b}_n r)$  としてもよい。 $b = (b_0, b_n)'$  を使って表わすと、

終価と年価

$$(3) \left\{ \begin{array}{l} S = \frac{1}{r} \{ (1+r)^n - (1+r) \} M + (1+r)^n M_0 b_0 + M_n b_n \\ M_0 = M_n = M \text{ のときのみ陽に表現でき、} \\ M = \frac{r}{(1+r)^{n+b_0} - (1+\bar{b}_n r)} S \end{array} \right.$$

現価と年価

$$(4) \left\{ \begin{array}{l} P = \frac{1}{r} \left\{ \frac{1}{(1+r)^n} - \frac{1}{1+r} \right\} M + M_0 b_0 + \frac{1}{(1+r)^n} M_n b_n \\ M_0 = M_n = M \text{ のときのみ陽に表現でき、} \\ M = \frac{r(1+r)^n}{(1+r)^{n+b_0} - (1+\bar{b}_n r)} P \end{array} \right.$$

なお、 $M_0 = M_n = M$  でないときは次回、年価・金利が一定でないときの公式を使用すればよい。

以上の式をふまえて  $J$  で関数を作成した。関数名は求める項目名が先頭にきて、関数名と引数は年利  $I$ 、期間  $N$ 、終価  $S$ 、原価  $P$ 、年価  $M$ 、1 年での金利計算回数  $Fre$ 、の順番に並べ、 $M$  があるときには 2 項関数で左引数に期末・期首などのオプションをつけることにした。

NB. =====

NB. 財務計算-基本関数 2011.02.17  
 NB. (期間と金利の陽関数を除く)  
 NB. i 利率 Intrest INT  
 NB. n 期間 Period PER  
 NB. P 現価 Present Value PV  
 NB. S 終価 Future Value FV  
 NB. M 年価 Payment PMT  
 NB. Fre 年複利回数 Frequency 1,2,3,4,6,12  
 NB. Left Arguments  
 NB. option OPT  
 NB. 0 期末 Arrears ARR  
 NB. 1 期首 Advance ADV  
 NB. 2 無無  
 NB. 3 有有  
 NB. 使い方 SINP i;n;P  
 NB. PINS i;n;S  
 NB. opt MINS i;n;S;Fre  
 NB. opt SINM i;n;M;Fre  
 NB. opt MINP i;n;P;Fre  
 NB. opt PINM i;n;M;Fre  
 NB.  
 NB. 例 1.100 万円を金利 7%で 10 年間据え置いたとき満期での元利合計金額  
 NB. SINP 0.07;10;100 答: 約 196 万 7000 円  
 NB. 例 2. 金利 7%で 10 年間で 100 万円にするにはいくら預けたらいいか  
 NB. PINS 0.07;10;100 答: 50 万 8349 円  
 NB. 例 3. 金利 7%で 10 年間積立て満期で 100 万円受け取るための毎月積立金  
 NB. 0 MINS 0.07;10;100;12 答: 期末で解くと、5778 円  
 NB. 例 4.1000 万円を借り、金利 7%、期首、10 年間、月払いで返済  
 NB. 1 MINP 0.07;10;1000;12 答: 11 万 5435 円  
 NB. 例 5.1000 万円を借り、金利 7%、期首、10 年間、年払いでの返済額  
 NB. 1 MINP 0.07;10;1000;1 答: 133 万 0631 円  
 NB. 例 6. 金利 3%、10 年間、毎月 4 万円積立てると満期で幾ら貯まるか  
 NB. 0 SINM 0.03;10;4;12 答: 期末で解くと、558 万 9657 円  
 NB. 例 7. 金利 3%、10 年間月払いで 4 万円返済するとすれば幾ら借りられるか  
 NB. 0 PINM 0.03;10;4;12 答: 期末で解くと、414 万 2470 円

SINP=:3 : 'p\*(>i)^n=.n[i=.i[''i n p''=.y'

PINS=:3 : 's\*(>i)^n=.n[i=.i[''i n s''=.y'

MINS=:3 : 0

0 MINS y

```

:
if. x>:4 do. '** x is over 4 **' return. end.
'b0 bn'=.2 2#:x{0 3 1 2
s%i%((>:i)^(n+b0))-(1+bn*i)[n=.n*f[i=.i%f['i n s f'=.y
)
    SINM=:3 : 0
0 SINM y
:
if. x>:4 do. '** x is over 4 **' return. end.
'b0 bn'=.2 2#:x{0 3 1 2
m%i%((>:i)^(n+b0))-(1+bn*i)[n=.n*f[i=.i%f['i n m f'=.y
)
    MINP=:3 : 0
0 MINP y
:
if. x>:4 do. '** x is over 4 **' return. end.
'b0 bn'=.2 2#:x{0 3 1 2
p*((>:i)^n)*i%((>:i)^(n+b0))-(1+bn*i)[n=.n*f[i=.i%f['i n p f'=.y
)
    PINM=:3 : 0
0 PINM y
:
if. x>:4 do. '** x is over 4 **' return. end.
'b0 bn'=.2 2#:x{0 3 1 2
m%((>:i)^n)*i%((>:i)^(n+b0))-(1+bn*i)[n=.n*f[i=.i%f['i n m f'=.y
)
NB.=====

```

### 3 . 年価・金利が一定の場合の期間と金利の公式

年価が入らなければ  $n$  や  $r$  が陽の関数として表現できる。

年価がないときの  $n$  と  $r$

$$(5) \quad n = \frac{\ln \frac{S}{P}}{\ln(1+r)}$$

$$(6) \quad r = \left(\frac{S}{P}\right)^{\frac{1}{n}} - 1$$

年価が入っても  $M_0 = M_n = M$  のときは  $n$  だけは陽の関数として表現できる。

終価と年価による  $n$

$$(7) \quad n = \ln \left\{ (1 + \bar{b}_n * r) + \frac{S}{M} r \right\} / \ln(1+r) - b_0$$

現価と年価による  $n$

$$(8) \quad n = \ln \left\{ \frac{1 + \bar{b}_n * r}{1 + b_0 * r - \frac{P}{M} r} \right\} / \ln(1 + r)$$

ただし、前述したように  $\bar{b}_n = \text{not } b_n$  である。

### 金利の公式

$M$  が入った場合、金利の式  $r$  に関しては残念ながら  $N$ 、 $S$ 、 $P$ 、 $M$  の陽関数で表わすことができない。そこで  $J$  の関数は (2) もしくは (3)・(4) の式を用いて  $r$  に対する二分探索法によって  $r$  を求めることにした。エクセルではゴールシークを使って求めることもできるし、後で述べる内部利率関数 IRR を使っても求めることもできる。

この関数 INSM と INPM 使い方は、関数の使い方の例 6、例 7 のところで示してある。

NB. =====

NB. 財務計算-基本計算 2      2011.02.17

NB. (期間と年利の式)

NB. P 現価 Present Value PV

NB. S 終価 Future Value FV

NB. M 年価 Payment            PMT

NB. i 利率 Intrest            INT

NB. n 期間 Period            PER

NB. Option                    OPT

NB. 0 期末 Arrears            ARR

NB. 1 期首 Advance            ADV

NB. 2 無無

NB. 3 有有

NB. 1~12 1,2,3,4,6,12 Fre 年複利回数

NB. 使い方      INPS n;P;S

NB.              NIPS i;n;S

NB.      opt NISM i;S;M;Fre

NB.      opt NIPM i;P;M;Fre

NB.      opt INSM n;S;M;Fre

NB.      opt INPM n;P;M;Fre

NB.

NB. 例 1. 100 万円を預け 10 年後に 200 万円になるための年利は？

NB.      INPS 10;100;200 答：約 7.177%

NB. 例 2. 100 万円を預け金利 7% で 200 万円になるには何年かかるか？

NB.      NIPS 0.07;100;200 答：10.245 年だから約 10 年と 3 ヶ月

NB. 例 3. 年利 7% で毎月期末 1 万 5000 円を預け何年後に 1000 万円となるか

NB.      0 NISM 0.07;1000;1.5;12 答：約 22.74 年

NB. 例 4. 年利 3% で 1000 万円借り毎月期末で 4 万円を返済するときの年数は

NB. 0 NIPM 0.03;1000;4;12 答：約 32.735 年  
 NB. 例 5. 年利 3%で 1000 万円借り毎年期末で 48 万円を返済するときの年数は  
 NB. 0 NIPM 0.03;1000;48;1 答：約 33.18 年  
 NB. 例 6. 毎月期末で 1 万 5000 円を預け 30 年後に 1000 万円となる年利は  
 NB. 0 INSM 30;1000;1.5;12 答：約 3.77%  
 NB. 例 7. 1000 万円借り毎月期末で 4 万円を 30 年間返済するときの年利は  
 NB. 0 INPM 30;1000;4;12 答：約 2.59%

```

INPS=:3 : 0
if. (#y)~:3 do. '** y must be 3 elements **' return. end.
<:1%~(S%P)^%N[N=.n['n P S'=.y
)
NIPS=:3 : 0
if. (#y)~:3 do. '** y must be 3 elements **' return. end.
(^.S%P)^.>:I=.i['i P S'=:y
)
NISM=: 4 : 0
if. (#y)~:4 do. '** y must be 4 elements **' return. end.
'b0 bn'=.2 2#:x{0 3 1 2[I=.i%f['i S M f'=.y
f%~((^.(>:bn*I)+S*I%M)^.(>:I))-b0
)
NIPM=: 4 : 0
if. (#y)~:4 do. '** y must be 4 elements **' return. end.
'b0 bn'=.2 2#:x{0 3 1 2[I=.i%f['i P M f'=.y
f%~(^.(>:bn*I)%(>:b0*I)-P*I%M)^.(>:I)
)
NB.===== 二分探索法により利率を求める =====
bisearch=:4 : 0
NB. F bisearch lower;higher
NB. F is increasing,and root must be between lower and higher
'y1 y2'=:y[i=.0
NB. To avoid Global variables, Add the extra arguments
if. (_4{.>0{x)='P' do. 'F x N P M'=.x else. 'F x N S M'=.x end.
label_L1.
if.30<i=.>:i do. goto_owari.end.
z=."F,'y0=-:+/y1,y2'
if.z>0 do.y2=.y0 else.y1=.y0 end.
if.0.0000000001>|z do.goto_owari.end.
goto_L1.
label_owari.

```

```
y0
)
```

```
MS_1=: 3 : 0
NB. Usage : (xx;n;Fre) MS_1 I
'Not Use a Monadic Form'
:
'b0 bn'=.2 2#:xx{0 3 1 2[N=.n*f['xx n f'=.x
if. y=0 do. %N+(1*b0)+(_1*bn) return. end.
y%((>:y)^(N+b0))-(1+bn*y)
)
```

```
MP_1=: 3 : 0
NB. Usage : (xx;n;Fre) MP_1 I
'Not Use a Monadic Form'
:
N=.n*f['xx n f'=.x
((>:y)^N)*(xx;N;1)MS_1 y
)
```

```
INSM=:4 : 0 NB. 積立金利
NB. INSM N;S;M;Fre
if. (#y)~:4 do. 'members of y is not 4' return. end.
N=.n*f['n S M f'=.y
NB.if. S<:N*M do. 'S is less than total payments !!' return. end.
NB. To avoid Global variables, Add the extra arguments x;N;S;M
f*('M-S*(x;N;1)MS_1 ';x;N;S;M) bisearch _1;1
)
```

```
INPM=:4 : 0 NB. 内部金利 (IRR)、投資限界金利
NB. INPM N;P;M;Fre
if. (#y)~:4 do. 'members of y is not 4' return. end.
N=.n*f['n P M f'=.y
NB.if. P>N*M do. 'P is greater than total earnest !!' return. end.
NB. To avoid Global variables, Add the extra arguments x;N;P;M
f*('M~P*(x;N;1)MP_1 ';x;N;P;M) bisearch _1;1
)
NB.=====
```

#### 4 . 例題

本稿では関数のところで全関数に亘って詳細な例題が載っているので省略する。

## 【参考文献】

- 【1】 Office 2003 Microsoft Excel ( 2003 ): HELP 財務計算、Excel 内部関数資料
- 【2】 J601 システムパッケージ (1994-2006) : ファイナンスパッケージ (財務計算)、  
j601/system/packages/finance/interest.ijs、j601\_win.exe を解凍すると得られる
- 【3】 竹内寿一郎 ( 2010 ): 財務計算あれこれ 第 1 回 \_\_現価、終価、年価\_\_、 JAPLA 研  
究会 2010.1.23 資料
- 【4】 竹内寿一郎 ( 2010 ): 財務計算あれこれ 第 2 回 \_\_期間と利率の計算式\_\_、 JAPLA  
研究会 2010.2.27 資料
- 【5】 竹内寿一郎 ( 2010 ): 財務計算あれこれ 第 3 回 \_\_内部利率の計算式\_\_、 JAPLA 研  
究会 2010.3.27 資料
- 【6】 竹内寿一郎 ( 2010 ): 財務計算あれこれ 第 4 回 \_\_年率および年価が一定でない場  
合の計算式 (1) \_\_、 JAPLA 研究会 2010.10.23 資料
- 【7】 竹内寿一郎 ( 2010 ): 財務計算あれこれ 第 5 回 \_\_年率、年価および期間が一定で  
ない場合の計算式 (2) \_\_、 JAPLA2010 シンポジウム 於：立川統計数理研究所  
2010.12.04 シンポジウム資料
- 【8】 竹内寿一郎 ( 2011 ): 財務計算あれこれ 第 6 回 \_\_財務計算における色々な例題  
(1) \_\_、 JAPLA 研究会 2010.01.22 資料
- 【9】 竹内寿一郎 ( 2010 ): 万年暦「寿暦」の作成、 JAPLA 研究会 2010.11.27 資料